

Spivey's BirthdayBook Example in HOL-Z

June 16, 2008

Contents

1	The Specification	2
1.1	Basic Datatypes and State Schemas	2
1.2	Operations	2
1.3	Strengthening the Specification	2
1.4	Implementing the Birthday Book	3
1.5	Specification	3
1.6	Stating Conjectures	3
1.7	Conclusion	4
2	Import, Inspection, and Basic Analysis of the Specification	4
2.1	Loading a ZETA-unit and the Consequences	4
2.2	Analysis I: Proving Conjectures	5
2.3	Analysis II: Checking Consistency	5
3	Analysis by Functional Refinement	7
3.1	Setting up the functional refinement	7
3.2	Proof of Functionality of the Abstraction Relation	8
3.3	Proofs of the Init-Condition of the Refinement	8
3.4	Proof of the First Operation-Refinement-Condition	9
3.5	Proof of the Second Operation-Refinement-Condition	10
3.6	Global Checks	13
4	Analysis by Data (Relational) Refinement	14
4.1	Setting up the Relational Refinement	14
4.2	Proofs of the Init-Condition of the Refinement	14
4.3	Proof of the First Operation-Refinement-Condition	15
4.4	Proof of the Second Operation-Refinement-Condition	16
4.5	Global Checks	19
5	Include both Refinement Proof Alternatives	19

1 The Specification

This document contains specification and analysis of Spivey's classical BirthdayBook example. It is intended to demonstrate the use of the ZETA-frontend to write and typecheck specifications in Z, and the Isabelle backend that allows for stating and proving proof obligations over this specification.

1.1 Basic Datatypes and State Schemas

section *BBSpec*

$[NAME, DATE]$

<i>BirthdayBook</i>
$known : \mathbb{P} NAME$
$birthday : NAME \leftrightarrow DATE$
$known = \text{dom } birthday$

1.2 Operations

$InitBirthdayBook == [BirthdayBook \mid known = \emptyset]$

<i>AddBirthday</i>
$\Delta BirthdayBook$
$name? : NAME; date? : DATE$
$name? \notin known$
$birthday' = birthday \cup \{name? \mapsto date?\}$

<i>FindBirthday</i>
$\exists BirthdayBook$
$name? : NAME; date! : DATE$
$name? \in known$
$date! = birthday(name?)$

<i>Remind</i>
$\exists BirthdayBook$
$today? : DATE; cards! : \mathbb{P} NAME$
$cards! = \{n : NAME \mid n \in known; birthday(n) = today?\}$

1.3 Strengthening the Specification

$REPORT ::= ok \mid already_known \mid not_known$

<i>Success</i>
$result! : REPORT$
$result! = ok$

$\text{--- } \textit{AlreadyKnown} \text{ ---}$ $\exists \textit{BirthdayBook}$ $\textit{name?} : \textit{NAME}$ $\textit{result!} : \textit{REPORT}$ <hr/> $\textit{name?} \in \textit{known}$ $\textit{result!} = \textit{already_known}$
--

$R\textit{AddBirthday} == (\textit{AddBirthday} \wedge \textit{Success}) \vee \textit{AlreadyKnown}$

1.4 Implementing the Birthday Book

$\text{--- } \textit{BirthdayBook1} \text{ ---}$ $\textit{names} : \mathbb{N} \rightarrow \textit{NAME}$ $\textit{dates} : \mathbb{N} \rightarrow \textit{DATE}$ $\textit{hwm} : \mathbb{N}$ <hr/> $\forall i, j : 1 \dots \textit{hwm} \bullet i \neq j \Rightarrow \textit{names}(i) \neq \textit{names}(j)$
--

$\textit{InitBirthdayBook1} == [\textit{BirthdayBook1} \mid \textit{hwm} = 0]$

$\text{--- } \textit{Abs} \text{ ---}$ $\textit{BirthdayBook}$ $\textit{BirthdayBook1}$ <hr/> $\textit{known} = \{i : 1 \dots \textit{hwm} \bullet \textit{names}(i)\}$ $\forall i : 1 \dots \textit{hwm} \bullet \textit{birthday}(\textit{names}(i)) = \textit{dates}(i)$

$\text{--- } \textit{AddBirthday1} \text{ ---}$ $\Delta \textit{BirthdayBook1}$ $\textit{name?} : \textit{NAME}$ $\textit{date?} : \textit{DATE}$ <hr/> $\forall i : 1 \dots \textit{hwm} \bullet \textit{name?} \neq \textit{names}(i)$ $\textit{hwm}' = \textit{hwm} + 1$ $\textit{names}' = \textit{names} \oplus \{\textit{hwm}' \mapsto \textit{name?}\}$ $\textit{dates}' = \textit{dates} \oplus \{\textit{hwm}' \mapsto \textit{date?}\}$

1.5 Specification

1.6 Stating Conjectures

At times, the designer of a specification might want to state a certain property that he has in mind when writing the specification document. Such properties can be stated as

conjecture. ZETA can type-check them and export them to HOL-Z; the latter will consider a conjecture as definition of an internal constant symbol. For proving the conjecture, one states a lemma in the analysis that this internal constant symbol is actually equivalent to *True*.

The statement of a conjecture is simply done by:

$$\begin{aligned} & \forall \textit{BirthdayBook}; \textit{BirthdayBook1}; \textit{name?} : \textit{NAME}; \textit{date?} : \textit{DATE} \bullet \\ & \textit{name?} \notin \textit{known} \wedge (\textit{known} = \{i : 1..hwm \bullet \textit{names}(i)\}) \\ & \Rightarrow (\forall i : 1..hwm \bullet \textit{name?} \neq \textit{names}(i)) \end{aligned}$$

1.7 Conclusion

In the following, we discuss two versions of analysis: one based on relational refinement, another on functional refinement, which leads to simpler proof. See corresponding `BBSpec.thy` file for the relational refinement, and `BBSpec_Functional.thy` for the functional refinement.

The following ISAR command starts a new Isabelle theory based on `Z`, including all libraries and setups.

2 Import, Inspection, and Basic Analysis of the Specification

```
theory BBSpec
imports Z
```

```
begin
```

2.1 Loading a ZETA-unit and the Consequences

The following HOL-Z toplevel commands allows for loading directly the output of the ZETA format, in this case, we load the previous specification `BBSpec` presented in the previous section.

```
ML{* print_mode := !print_mode @ ["xsymbols"]; *}
```

```
load_holz "BBSpec"
```

```
thm AddBirthday_def
```

This leads to a new state of the proof environment where all sorts of elements of the specification were bound to ISAR names and can therefore be referenced in future proofs.

A guided tour through the generated definitions looks as follows: For abstract types, constants describing the type sets exist:

```
thm NAME_def
```

Data types result in a number of simplification rules

```
thm BBSpec.REPORT.simps
```

Schema (both states and operation schemas) were converted to constant definitions denoting the "set of records", i.e. their value

```
thm BirthdayBook_def

thm Remind_def
  InitBirthdayBook_def
  AddBirthday_def
  RAddBirthday_def
```

Schemas, axiomatic definitions and conjectures were collected into the variables:

```
thm SCHEMAS
thm AXDEFS
thm CONJECTURES
```

2.2 Analysis I: Proving Conjectures

Example:

```
lemma conjecture_0_proof : "conjecture_0"
by(unfold conjecture_0_def,zstrip,
  unfold BirthdayBook_def BirthdayBook1_def, auto simp: Z2HOL)
```

Of course, the conjecture can also be stated in the analysis document (the .thy-file) directly.

```
zlemma PO_refine_1_AddBirthday_simple :
"  $\forall$  BirthdayBook  $\bullet$  ( $\forall$  BirthdayBook1  $\bullet$ 
  ( $\forall$  name?  $\in$  NAME.  $\forall$  date?  $\in$  DATE.
    ((name?  $\notin$  known  $\wedge$  known = {n.  $\exists$  i  $\in$  1..hwm. n = names(.i.)})
     $\longrightarrow$ 
    ( $\forall$  i  $\in$  1..hwm. name?  $\neq$  names(.i.)))
  )"
```

```
by(zstrip,auto)
```

Note that the input of the formula must be done with the zlemma command since this supports HOL-Z syntax (and not just HOL syntax).

2.3 Analysis II: Checking Consistency

Now we turn to the statement of analysis judgements and refinement judgements. Both lead to the generation of proof obligations.

The refinement package provides a number of generic toplevel commands for these analysis judgements and, moreover, methods to inspect and discharge proof obligations.

Such method-specific support helps a lot to improve critical review of specifications ("is this really what you want to specify?"), as well as a basis for specific tactics.

Generic inspection commands are:

```
list_po
check_po
```

... which provide functionality for listing pending proof obligations and also checks that they have been discharged. At the moment, both commands are nops since no proof obligation has been generated.

The first analytic judgement statement commands are:

```
gen_state_cc BirthdayBook
gen_state_cc BirthdayBook1
```

... which lead to the generation of the proof obligations `BBSpec.ccState_BirthdayBook_1` and `BBSpec.ccState_BirthdayBook1_1`. They state that the state schemas (representing invariants) should be satisfiable. With the generic command:

```
show_po      BBSpec.ccState_BirthdayBook1_1
```

As an example, we will discharge this proof obligation by providing a proof for it.

```
po "BBSpec.ccState_BirthdayBook1_1"
```

This means that we have to show:

1. \exists `BirthdayBook1` • `True`

Obviously, we have to eliminate the schema quantifier by an respective introduction tactic for the schema calculus:

```
apply (zintro_sch_ex,clarify,(rule refl)+)
apply (unfold BirthdayBook1_def)
apply (auto simp: Z2HOL)
apply (rule_tac [3] ZInteg.zero_is_natural, simp_all)
apply (rule_tac f="λ x. arbitrary" in lambda_total1,simp)+
discharged
```

... we can display the precise form of proof obligation.

Moreover, we generate proof obligations that assure that an operation schema is in fact implementable or “non-blocking”, in the sense that there is a (not necessarily computable) function mapping pre-state and input to a post-state and output.

```
gen_op_cc      AddBirthday
gen_op_cc      AddBirthday1
```

```
show_po      BBSpec.ccOp_AddBirthday1_1
```

Note that the listing of now active proof obligations can be parameterized by filters excluding certain proof-obligation classes (more filtering functions are desirable, but currently not implemented):

```
list_po except ccOp
```

```
end
```

3 Analysis by Functional Refinement

```
theory Fun_Refinement
imports BBSpec
```

```
begin
```

3.1 Setting up the functional refinement

Now we set the default abstraction relation in the refinement package; this setting is a pre-requisite to the future generation of refinement related proof obligations.

```
set_abs "Abs"[functional]
```

At this place, we add the directive "[functional]" in order to indicate that the abstraction relation is in fact a function.

This results in the additional proof obligation that the abstraction relation is in fact a function, but leads to simpler proof obligations over this abstraction relation later.

Now comes the core of proof obligation generation based on the Forward-Simulation Refinement Method for Z (see Spivey's Book or "Using Z")

```
refine_init InitBirthdayBook InitBirthdayBook1
refine_op   AddBirthday       AddBirthday1
```

```
show_po Fun_Refinement.fwRefinementInit_BirthdayBook_1
show_po Fun_Refinement.fwRefinementOp_AddBirthday_2
list_po
```

```
show_po BBSpec.ccOp_AddBirthday_1
        BBSpec.ccOp_AddBirthday1_1
        BBSpec.ccState_BirthdayBook_1
        BBSpec.ccState_BirthdayBook1_1
        Fun_Refinement.fwRefinementFunctional_Abs_1
        Fun_Refinement.fwRefinementOp_AddBirthday_1
        Fun_Refinement.fwRefinementOp_AddBirthday_2
        Fun_Refinement.fwRefinementInit_BirthdayBook_1
```

We perform a final check of the proof obligations; however, we filter out certain classes of proof-obligations.

```
check_po except ccOp ccState fwRefinementFunctional
              fwRefinementOp fwRefinementInit
```

In this example, we actually filter out **all** obligations.

In contrast, the toplevel command:

```
check_po
```

would result in a failure:

```
*** There are 7 unproven proof-obligations (can not ignore!).
*** Check failed.
*** At command "check_po".
```

3.2 Proof of Functionality of the Abstraction Relation

lemma lemma1:

```
"[[ BirthdayBook1 (dates, hwm, names); i ∈ ( 1 .. hwm); ia ∈ ( 1 .. hwm);
  names(.i.) = names(.ia.) ]]
  ⇒ dates(.i.) = dates(.ia.)"
apply (unfold BirthdayBook1_def, simp add: Z2HOL, clarify)
apply (case_tac "i=ia", auto)
done
```

lemma lemma2:

```
" (a ∈ (rel_appl names) ' (1 .. hwm)) = (∃ i ∈ 1 .. hwm. names(.i.) = a)"
apply (auto simp: Z2HOL)
done
```

```
po "Fun_Refinement.fwRefinementFunctional_Abs_1"
apply (zstrip)
apply (simp add: Z2HOL Ex1_def)
apply (rule_tac x="{(x,y). ∃ i ∈ (1 .. hwm). x = names ^ i ∧ y = dates ^ i}" in exI)
apply (rule_tac x="(rel_appl names) ' (asSet(λi. i : ( 1 .. hwm)))" in exI)
apply (unfold Abs_def BirthdayBook_def)
apply (simp add: Z2HOL Ex1_def)
apply (safe, simp_all)
apply (simp only: pfun_def rel_def, auto intro!: lemma1)+
apply (subst ZFun.beta_apply_pfun[of _ NAME DATE])
prefer 3
apply (rule refl)
apply (auto)
apply (rule pfunI)
apply (simp add: rel_def)
apply auto
apply (auto intro!: lemma1)
prefer 2
apply (rule_tac t="dates ^ i" in subst)
prefer 2
apply (erule ZFun.rel_apply_in_rel, auto)
apply (drule_tac x=aa in eqset_imp_iff, auto)
apply (rule_tac x=x in bexI, auto)
discharged
```

3.3 Proofs of the Init-Condition of the Refinement

po Fun_Refinement.fwRefinementInit_BirthdayBook_1

To show:

1. $\forall \text{ BirthdayBook} \bullet (\forall \text{ BirthdayBook1} \bullet (\text{InitBirthdayBook1} \wedge \text{Abs} \longrightarrow \text{InitBirthdayBook}))$

We perform structural simplification by eliminating Schema-Calculus-Constructs.

apply zstrip

Now we follow the brute force approach: unfolding all schema definitions ...

```
apply (unfold InitBirthdayBook1_def InitBirthdayBook_def Abs_def BirthdayBook_def)
```


Conversion to plain HOL and using the simplifier just finds the witness $(\{\}, \{\})$ automatically.

```
apply(simp add: Z2HOL)
discharged
```

3.4 Proof of the First Operation-Refinement-Condition

In the following, we introduce three lemmas that allow the reduction of the first refinement condition to the simplified version above.

We provide three auxilliary lemmas.

First: the syntactic precondition over leagal states implies the semantic precondition for AddBirthday1:

```
zlemma lemma3 :
" BirthdayBook1  $\wedge$  ( $\forall$  i $\in$ 1..hwm. name?  $\neq$  (names(i.)))  $\longrightarrow$  pre AddBirthday1"
```

We start with elementary Z-logical massage:

```
apply(zstrip, zintro_pre AddBirthday1_def)
apply(simp add: DECL_def DELTA_def, rule conjI)
```

...and split the declarations from body.

In particular, we take the prescribed successor state and propagate it in proof.

```
apply(rule_tac [2] conjI | rule_tac [2] refl)+
```

...proves that the successor state fulfills state invariant.

```
apply(zunfold BirthdayBook1_def)
apply(simp add: Ball_def maplet_def zpred_def, auto)
```

...does its best to make it simpler

auto reduces the proof to a "proof by contradiction" scheme ... Mostly, it has to do with

$$\begin{aligned} &(\text{names } (+) \{(\text{hwm} + 1, \text{name?})\}) \%^{\wedge} i = \\ &(\text{names } (+) \{(\text{hwm} + 1, \text{name?})\}) \%^{\wedge} j; \end{aligned}$$

while we know that $\text{names } \%^{\wedge} i \sim = \text{names } \%^{\wedge} j$. We bring this goal in the end of the assumption list:

```
apply(rotate_tac 1)
```

Now comes the proof idea: We split up a case-distinction tree for the cases that i and j refer to the new element ... rotate_tac brings these clauses in front in the assumption list and makes them visible for the rewriter.

```
apply(case_tac [1] "x=(hwm+1)")
apply(case_tac [1] "xa=(hwm+1)")
apply(case_tac [3] "xa=(hwm+1)")
```

Since we know already that $i \sim = j$ and ALL $i: \#1 .. \text{hwm}$. $\text{name?} \sim = \text{names } \%^{\wedge} i$, these four cases can be reduced ad absurdum.

```
apply(auto simp: zpred_def)
done
```

Second: The semantic precondition of the abstract operation implies syntactic precondition

```

zlemma lemma4 : "pre AddBirthday  $\longrightarrow$  name?  $\notin$  known"
by(zstrip,zelim_pre,zunfold AddBirthday_def,auto)

```

here comes a structural proof for the first main goal: use the above three Z-lemmas in order to reduce the main goal to the simplified version of it above. The technique applies the stripS converter to bring Z-lemmas on-thy-fly into HOL form and introduce them into the proof by Isabelle standard tactics.

```

po Fun_Refinement.fwRefinementOp_AddBirthday_1

```

To show:

```

  1.  $\forall$  BirthdayBook  $\bullet$  ( $\forall$  BirthdayBook1  $\bullet$  ( $\forall$  date? name?. pre AddBirthday  $\wedge$  Abs  $\longrightarrow$ 
pre AddBirthday1))

```

After structural normalization:

```

apply(zstrip, clarify)

```

...we use the lemmas 1) to 3) by weakening assumptions and reducing conclusions.

```

apply(zrule lemma3,zdrule lemma4,zdrule Abs_def[zpred [1]])
apply(auto simp: Z2HOL)
discharged

```

3.5 Proof of the Second Operation-Refinement-Condition

To establish the second refinement condition, we need two auxilliary lemmas:

lemma lemma6:

```

" BirthdayBook (birthday, known)  $\implies$  birthday : NAME  $\leftrightarrow$  DATE"
by(zstrip, unfold BirthdayBook_def,simp add: Z2HOL)

```

Notice that you actually *can* use the ' symbol in the lexis; However, this works in the standard HOL parser and has nothing to do with the Z stroke convention ...

lemma lemma7:

```

"[[ BirthdayBook (birthday, {a.  $\exists i \in 1 .. hwm$ . names(.i) = a});
  BirthdayBook (birthday'a, {a.  $\exists i \in 1 .. hwm'$ . names'(.i) = a});
  BirthdayBook1 (dates, hwm, names); BirthdayBook1 (dates', hwm', names');
  AddBirthday1 (dateI, dates, dates', hwm, hwm', nameI, names, names');
  BirthdayBook (birthday', known');
   $\forall i \in 1 .. hwm$ . names(.i)  $\neq$  nameI;
  birthday' = insert (nameI, dateI) birthday;
   $\forall i \in 1 .. hwm$ . birthday % $\wedge$  (names(.i)) = dates(.i);
   $\forall i \in 1 .. hwm'$ . birthday'a % $\wedge$  (names'(.i)) = dates'(.i) ]]
 $\implies$  dom birthday'a = insert nameI (dom birthday)"

```

```

apply(subgoal_tac "nameI  $\sim$ : dom birthday")
apply(simp add: insert_is_pfun)
apply(unfold BirthdayBook_def,simp add: Z2HOL)
apply((erule conjE)+, drule sym, simp)
apply(unfold AddBirthday1_def BirthdayBook1_def,simp add: maplet_def Z2HOL,clarify)
defer 1
apply(unfold BirthdayBook_def,simp add: maplet_def Z2HOL,clarify)
apply(drule_tac t="dom ?Z" in sym)+
apply(simp,blast)

```

we reorient the two crucial equalities in the assumptions:

```

apply(thin_tac "?X")
apply(drule_tac t="dom ?Z" in sym)+
apply(rule set_ext, simp,safe,simp)
apply(case_tac "i=hwm+1",simp)
apply(rotate_tac -2)
apply(erule_tac x=i in ballE,simp)

... which yields a contradiction

apply(simp add: zpred_def)
apply(rule_tac x="hwm+1" in bexI,simp, simp,
      simp add: numb_range_def in_naturals[symmetric])
apply(rule_tac x=i in bexI)
apply(thin_tac "ALL x:?S. ?P x")+
apply(thin_tac "?T = ?U")+
apply(subst oplus_by_pair_apply2, simp add: numb_range_def,simp)
apply(simp add: numb_range_def)
done

```

lemma lemma8:

```

"[[ BirthdayBook (birthday, {a.  $\exists i \in 1 .. hwm. \text{names } i = a$ });
  BirthdayBook (birthday'a, {a.  $\exists i \in 1 .. hwm. \text{names}' i = a$ });
  BirthdayBook1 (dates, hwm, names); BirthdayBook1 (dates', hwm', names');
  AddBirthday1 (dateI, dates, dates', hwm, hwm', nameI, names, names');
  BirthdayBook (birthday', known');
   $\forall i \in 1 .. hwm. \text{names } i \neq \text{nameI}$ ;
  birthday' = insert (nameI, dateI) birthday;
   $\forall i \in 1 .. hwm. \text{birthday } i = \text{dates } i$ ;
   $\forall i \in 1 .. hwm'. \text{birthday}' a i = \text{dates}' i$ ;
   $i \in \text{dom birthday}' a; \text{dom birthday}' a = \text{dom (insert (nameI, dateI) birthday)}$  ]]
 $\implies \text{birthday}' a(i) = \text{insert (nameI, dateI) birthday}(i)$ "
apply(subgoal_tac "nameI ~: dom birthday")
apply(simp add: insert_is_pfun)
apply(zunfold BirthdayBook_def,simp add: Z2HOL)
apply((erule conjE)+, drule sym, simp)
apply(zunfold AddBirthday1_def,zunfold BirthdayBook1_def,simp add: maplet_def Z2HOL,clarify)
defer 1
apply(zunfold BirthdayBook_def,simp add: maplet_def Z2HOL,clarify)
apply(drule_tac t="dom ?Z" in sym)+
apply(simp,blast)
apply(erule disjE, simp, (thin_tac "?T<:?S = ?U")+

```

Case A: i is equal to $hwm + 1$. Tis boils down to:

```

1.  $\forall i \in 1 .. hwm + 1. \forall j \in 1 .. hwm + 1. i \neq j \implies (\text{names} \oplus \{(hwm + 1, \text{nameI})\})(i) \neq (\text{names} \oplus \{(hwm + 1, \text{nameI})\})(j); \forall i \in 1 .. hwm. \text{names}(i) \neq \text{nameI}; \forall i \in 1 .. hwm. \text{birthday}(\text{names}(i)) = \text{dates}(i); \forall i \in 1 .. hwm + 1. \text{birthday}' a((\text{names} \oplus \{(hwm + 1, \text{nameI})\})(i)) = (\text{dates} \oplus \{(hwm + 1, \text{dateI})\})(i); \text{dom birthday}' a = \text{insert nameI (dom birthday)}; \text{nameI} \notin \text{dom birthday}; \text{birthday} \in \text{NAME} \leftrightarrow \text{DATE}; \{a. \exists i \in 1 .. hwm. \text{names}(i) = a\} = \text{dom birthday}; \text{birthday}' a \in \text{NAME} \leftrightarrow \text{DATE}; \{a. \exists i \in 1 .. hwm. (\text{names} \oplus \{(hwm + 1, \text{nameI})\})(i) = a\} = \text{insert nameI (dom birthday)}; \text{names} \in \mathbb{N} \rightarrow \text{NAME}; \forall i \in 1 .. hwm. \text{nameI} \neq \text{names}(i); \text{dates} \in \mathbb{N} \rightarrow \text{DATE}; hwm \in \mathbb{N}; \forall i \in 1 .. hwm. \forall j \in 1 .. hwm. i \neq j \implies \text{names}(i) \neq \text{names}(j); i = \text{nameI}]] \implies \text{birthday}' a(\text{nameI}) = \text{dateI}$ 
```

```

apply (drule_tac x="nameI" in eqset_imp_iff)
apply (simp, safe, thin_tac "?X")
apply (erule_tac x="hwm+1" and A="1..hwm+1" in ballE, simp)
apply (simp add: numb_range_def in_naturals[symmetric])

```

Case B: i is less to $hwm + 1$, i.e. in the domain of *birthdaybook*:

1. $\bigwedge y. \llbracket \forall i \in 1 .. hwm + 1. \forall j \in 1 .. hwm + 1. i \neq j \longrightarrow (names \oplus \{(hwm + 1, nameI)\})(.i) \neq (names \oplus \{(hwm + 1, nameI)\})(.j); \forall i \in 1 .. hwm. names(.i) \neq nameI; \forall i \in 1 .. hwm. birthday(.i) = dates(.i); \forall i \in 1 .. hwm + 1. birthday'a((names \oplus \{(hwm + 1, nameI)\})(.i)) = (dates \oplus \{(hwm + 1, dateI)\})(.i); \text{dom } birthday'a = \text{insert } nameI (\text{dom } birthday); nameI \notin \text{dom } birthday; NAME \triangleleft birthday = birthday; birthday \in NAME \leftrightarrow DATE; \{a. \exists i \in 1 .. hwm. names(.i) = a\} = \text{dom } birthday; birthday'a \in NAME \leftrightarrow DATE; \{a. \exists i \in 1 .. hwm. (names \oplus \{(hwm + 1, nameI)\})(.i) = a\} = \text{insert } nameI (\text{dom } birthday); names \in \mathbb{N} \rightarrow NAME; \forall i \in 1 .. hwm. nameI \neq names(.i); dates \in \mathbb{N} \rightarrow DATE; hwm \in \mathbb{N}; \forall i \in 1 .. hwm. \forall j \in 1 .. hwm. i \neq j \longrightarrow names(.i) \neq names(.j); (i, y) \in birthday \rrbracket \implies birthday'a(.i) = \text{insert } (nameI, dateI) birthday(.i) \rrbracket$

```

apply (thin_tac "?X", (thin_tac "?T < ?S = ?U")+ )
apply (drule_tac x=i in eqset_imp_iff) back
apply (simp, safe)
apply (rule_tac A="1 .. hwm+1" and x = i in ballE)
apply assumption
apply (subgoal_tac "(names \oplus \{(hwm + 1, nameI)\})(.i) = names(.i)", simp)
apply (subst oplus_by_pair_apply2)
apply (simp add: numb_range_def in_naturals[symmetric])
apply (rotate_tac 1)
apply (erule_tac x=i in ballE, simp, simp)
apply (subst oplus_by_pair_apply2)

```

The proof conclusion is somewhat messy. The proof state is so cluttered up with facts, that it takes the automated procedures quite some time to find the contradiction here. Thinning the assumption lists therefore greatly improves the proof speed.

```

apply ((thin_tac "ALL x:?S. ?P x")+ , (thin_tac "?T = ?U")+ , (thin_tac "?X : ?Y \rightarrow ?Z")+ ,
  simp add: numb_range_def in_naturals[symmetric], simp)
apply ((thin_tac "ALL x:?S. ?P x")+ , (thin_tac "?T = ?U")+ , (thin_tac "?X : ?Y \rightarrow ?Z")+ ,
  thin_tac "?X", thin_tac "?X", thin_tac "?X", thin_tac "?X", thin_tac "?X")
apply (simp add: numb_range_def in_naturals[symmetric])
done

```

po *Fun_Refinement.fwRefinementOp_AddBirthday_2*

To show:

1. $\forall BirthdayBook \bullet (\forall BirthdayBook \bullet (\forall BirthdayBook1 \bullet (\forall BirthdayBook1 \bullet (\forall date? name?. \text{pre } AddBirthday \wedge Abs \wedge AddBirthday1 \wedge Abs' \longrightarrow AddBirthday))))$

After structural simplification:

```

apply (zstrip, clarify, zelim_pre)
apply (unfold Abs_def AddBirthday_def,
  simp_all add: Z2HOL rel_appl_norm maplet_def, clarify)

```

This leads to the proof-state presented in Spivey proof (pp. 141):

1. $\bigwedge birthday \text{ known } birthday'a \text{ known}' a \text{ dates } hwm \text{ names } dates' hwm' \text{ names}' date? name? birthday' \text{ known}' . \llbracket BirthdayBook (birthday'a, \{a. \exists i \in 1 .. hwm'. names'(.i) = a\}); BirthdayBook1$

```
(dates, hwm, names); BirthdayBook1 (dates', hwm', names'); AddBirthday1 (date?, dates,
dates', hwm, hwm', name?, names, names');  $\forall i \in 1 \dots hwm. \text{birthday}(.(\text{names}(.i)).) = \text{dates}(.i.);$ 
 $\forall i \in 1 \dots hwm'. \text{birthday}'a(.(\text{names}'(.i)).) = \text{dates}'(.i.);$  BirthdayBook (birthday, {a.  $\exists i \in 1 \dots hwm. \text{names}(.i.) = a$ }); BirthdayBook (insert (name?, date?) birthday, known');  $\forall i \in 1 \dots hwm. \text{names}(.i.) \neq \text{name?}$ ]  $\implies \text{birthday}'a = \text{insert} (\text{name?}, \text{date?}) \text{birthday}$ 
```

Now we apply, as suggested in pp. 14, the extensionality rule on partial functions:

```
apply(rule pfun_ext)
```

Extensionality works only under condition that both sides are in fact partial functions. Spiveys proof misses this detail.

```
apply(erule lemma6)+
```

Now we reach the mail case distinction shown on page 14: we have to show that both domains are equal, and that they are pointwise equal:

```
We first treat the domain equality: 1.  $\bigwedge \text{birthday known birthday}'a \text{ known}'a \text{ dates hwm}$ 
names dates' hwm' names' date? name? birthday' known'.  $\llbracket \text{BirthdayBook} (\text{birthday}'a, \{a. \exists i \in 1 \dots hwm'. \text{names}'(.i.) = a\}); \text{BirthdayBook1} (\text{dates}, \text{hwm}, \text{names}); \text{BirthdayBook1} (\text{dates}', \text{hwm}', \text{names}'); \text{AddBirthday1} (\text{date?}, \text{dates}, \text{dates}', \text{hwm}, \text{hwm}', \text{name?}, \text{names}, \text{names}'); \forall i \in 1 \dots hwm. \text{birthday}(.(\text{names}(.i)).) = \text{dates}(.i.); \forall i \in 1 \dots hwm'. \text{birthday}'a(.(\text{names}'(.i)).) = \text{dates}'(.i.); \text{BirthdayBook} (\text{birthday}, \{a. \exists i \in 1 \dots hwm. \text{names}(.i.) = a\}); \text{BirthdayBook} (\text{insert} (\text{name?}, \text{date?}) \text{birthday}, \text{known}'); \forall i \in 1 \dots hwm. \text{names}(.i.) \neq \text{name?} \rrbracket \implies \text{dom } \text{birthday}'a = \text{dom} (\text{insert} (\text{name?}, \text{date?}) \text{birthday})$ 
```

```
apply(simp no_asm))
```

```
apply(rule_tac birthday'="insert (name?, date?) birthday" in lemma7,simp_all)
```

```
Now comes the part with the pointwise argument: 1.  $\bigwedge \text{birthday birthday}'a \text{ dates hwm}$ 
names dates' hwm' names' date? name? known' i.  $\llbracket \text{BirthdayBook} (\text{birthday}'a, \{a. \exists i \in 1 \dots hwm'. \text{names}'(.i.) = a\}); \text{BirthdayBook1} (\text{dates}, \text{hwm}, \text{names}); \text{BirthdayBook1} (\text{dates}', \text{hwm}', \text{names}'); \text{AddBirthday1} (\text{date?}, \text{dates}, \text{dates}', \text{hwm}, \text{hwm}', \text{name?}, \text{names}, \text{names}'); \forall i \in 1 \dots hwm. \text{birthday}(.(\text{names}(.i)).) = \text{dates}(.i.); \forall i \in 1 \dots hwm'. \text{birthday}'a(.(\text{names}'(.i)).) = \text{dates}'(.i.); \text{BirthdayBook} (\text{birthday}, \{a. \exists i \in 1 \dots hwm. \text{names}(.i.) = a\}); \text{BirthdayBook} (\text{insert} (\text{name?}, \text{date?}) \text{birthday}, \text{known}'); \forall i \in 1 \dots hwm. \text{names}(.i.) \neq \text{name?}; i = \text{name?} \vee i \in \text{dom } \text{birthday}; \text{dom } \text{birthday}'a = \text{insert } \text{name?} (\text{dom } \text{birthday}) \rrbracket \implies \text{birthday}'a(.i.) = \text{insert} (\text{name?}, \text{date?}) \text{birthday}(.i.)$ 
```

```
apply(rule_tac birthday'a="birthday'a" and birthday'="insert (name?, date?) birthday"
```

```
in lemma8,simp_all)
```

```
discharged
```

3.6 Global Checks

Now we check that all refinement conditions have indeed been proven:

```
check_po except ccOp ccState
```

This concludes the refinement proof based on Forward Simulation in the style of Spivey for the Birthdaybook example.

```
end
```

4 Analysis by Data (Relational) Refinement

```
theory Rel_Refinement
imports BBSpec
```

```
begin
```

4.1 Setting up the Relational Refinement

Now we set the default abstraction relation in the refinement package; this setting is a pre-requisite to the future generation of refinement related proof obligations.

```
set_abs "Abs"
```

Now comes the core of proof obligation generation based on the Forward-Simulation Refinement Method for Z (see Spivey's Book or "Using Z")

```
refine_init InitBirthdayBook InitBirthdayBook1
refine_op   AddBirthday       AddBirthday1

show_po Rel_Refinement.fwRefinementInit_BirthdayBook_1
show_po Rel_Refinement.fwRefinementOp_AddBirthday_2
list_po

show_po BBSpec.ccOp_AddBirthday_1
       BBSpec.ccOp_AddBirthday1_1
       BBSpec.ccState_BirthdayBook_1
       BBSpec.ccState_BirthdayBook1_1
       Rel_Refinement.fwRefinementOp_AddBirthday_1
       Rel_Refinement.fwRefinementOp_AddBirthday_2
       Rel_Refinement.fwRefinementInit_BirthdayBook_1
```

We perform a final check of the proof obligations; however, we filter out certain classes of proof-obligations.

```
check_po except ccOp ccState fwRefinementFunctional
              fwRefinementOp fwRefinementInit
```

In this example, we actually filter out **all** obligations.

In contrast, the toplevel command:

```
check_po
```

would result in a failure:

```
*** There are 7 unproven proof-obligations (can not ignore!).
*** Check failed.
*** At command "check_po".
```

4.2 Proofs of the Init-Condition of the Refinement

```
po Rel_Refinement.fwRefinementInit_BirthdayBook_1
```

To show:

1. $\forall \text{ BirthdayBook1} \bullet (\text{InitBirthdayBook1} \longrightarrow \exists \text{ BirthdayBook} \bullet \text{Abs} \wedge \text{InitBirthdayBook})$

We perform structural simplification by eliminating Schema-Calculus-Constructs.

apply *zstrip*

Now we follow the brute force approach: unfolding all schema definitions ...

apply(*zunfold* *InitBirthdayBook1_def* *InitBirthdayBook_def*
Abs_def *BirthdayBook_def*)

Conversion to plain HOL and using the simplifier just finds the witness ($\{\}$, $\{\}$) automatically.

apply(*simp* *add: Z2HOL*)
discharged

4.3 Proof of the First Operation-Refinement-Condition

In the following, we introduce three lemmas that allow the reduction of the first refinement condition to the simplified version above.

We provide three auxiliary lemmas.

First: the syntactic precondition over legal states implies the semantic precondition for *AddBirthday1*:

zlemma *lemma1* :
"*BirthdayBook1* \wedge ($\forall i \in 1..hwm. \text{name?} \neq (\text{names}(i))$) \longrightarrow *pre AddBirthday1*"

We start with elementary Z-logical massage:

apply(*zstrip, zintro_pre AddBirthday1_def*)
apply(*simp* *add: DECL_def DELTA_def, rule conjI*)

...and split the declarations from body.

In particular, we take the prescribed successor state and propagate it in proof.

apply(*rule_tac [2] conjI | rule_tac [2] refl*)⁺

...proves that the successor state fulfills state invariant.

apply(*zunfold BirthdayBook1_def*)
apply(*simp* *add: Ball_def maplet_def zpred_def, auto*)

...does its best to make it simpler

auto reduces the proof to a "proof by contradiction" scheme ... Mostly, it has to do with

$$\begin{aligned} &(\text{names } (+) \{(\text{hwm} + 1, \text{name?})\}) \%^{\wedge} i = \\ &(\text{names } (+) \{(\text{hwm} + 1, \text{name?})\}) \%^{\wedge} j; \end{aligned}$$

while we know that $\text{names } \%^{\wedge} i \sim = \text{names } \%^{\wedge} j$. We bring this goal in the end of the assumption list:

apply(*rotate_tac 1*)

Now comes the proof idea: We split up a case-distinction tree for the cases that *i* and *j* refer to the new element ... *rotate_tac* brings these clauses in front in the assumption list and makes them visible for the rewriter.

apply(*case_tac [1] "x=(hwm+1)"*)
apply(*case_tac [1] "xa=(hwm+1)"*)

```
apply(case_tac [3] "xa=(hwm+1)")
```

Since we know already that $i \sim j$ and $\text{ALL } i: \#1 \dots \text{hwm}. \text{name? } \sim \text{names } \%^i$, these four cases can be reduced ad absurdum.

```
apply(auto simp: zpred_def)
done
```

Second: The semantic precondition of the abstract operation implies syntactic precondition

```
zlemma lemma2 : "pre AddBirthday  $\longrightarrow$  name?  $\notin$  known"
by(zstrip,zelim_pre,zunfold AddBirthday_def,auto)
```

here comes a structural proof for the first main goal: use the above three Z-lemmas in order to reduce the main goal to the simplified version of it above. The technique applies the stripS converter to bring Z-lemmas on-thy-fly into HOL form and introduce them into the proof by Isabelle standard tactics.

```
po Rel_Refinement.fwRefinementOp_AddBirthday_1
```

To show:

```
1.  $\forall$  BirthdayBook  $\bullet$  ( $\forall$  BirthdayBook1  $\bullet$  ( $\forall$  date? name?. pre AddBirthday  $\wedge$  Abs  $\longrightarrow$  pre AddBirthday1))
```

After structural normalization:

```
apply(zstrip, clarify)
```

...we use the lemmas 1) to 3) by weakening assumptions and reducing conclusions.

```
apply(zrule lemma1,zdrule lemma2,zdrule Abs_def[zpred [1]])
apply(auto simp: Z2HOL)
discharged
```

4.4 Proof of the Second Operation-Refinement-Condition

To establish the second refinement condition, we need two auxilliary lemmas:

```
zlemma lemma4:
" (AddBirthday  $\wedge$  Abs  $\wedge$  AddBirthday1)  $\longrightarrow$ 
  known' = {n.  $\exists$  i  $\in$  1 .. hwm'. n = names'  $\%^i$  }"
```

After structural normalization ...

```
apply(zstrip, clarify)
```

...we unfold the definitions:

```
apply(simp add: set_simps prod_simps Z2HOL AddBirthday_def AddBirthday1_def Abs_def
  BirthdayBook_def BirthdayBook1_def,clarify)
apply(simp add: Dom_Union image_def asSet_def maplet_def)
```

We apply our main idea: apply reasoning via set extensionality and split equivalence into both directions ...

```
apply(rule set_ext,simp,rule iffI)
```

Proof \implies :

```
apply(erule disjE)
```



```

apply(rule_tac x="hwm+1" in bexI)
apply(simp add: override_apply numb_range_mem in_naturals)
apply(rule numb_range_mem)
apply(simp_all add: in_naturals)
apply(erule bexE)
apply(rule_tac x=xa in bexI)
apply(subst override_apply2)
apply(simp_all add: numb_range_def)

```

Proof \Leftarrow :

```

apply(erule exE)
apply(case_tac "i=hwm+1", simp)
apply(rule disjI2)
apply(rule_tac x=i in exI)
apply auto
done

```

```

zlemma lemma5 :
" (AddBirthday  $\wedge$  Abs  $\wedge$  AddBirthday1)  $\longrightarrow$ 
  ( $\forall$  i $\in$ 1 .. hwm'. birthday'^(names'~i) = dates'~i)"

```

After structural simplification and unfolding definitions :

```

apply(zstrip, clarify)
apply(simp add: set_simps prod_simps Z2HOL AddBirthday_def AddBirthday1_def Abs_def
  BirthdayBook_def BirthdayBook1_def,clarify)
apply(simp add: Dom_Union image_def asSet_def maplet_def)

```

we proceed by case distinction: either the accessed element is the last element:

```

apply(case_tac "i=hwm+1", auto)

```

...or it is before.

```

apply(subst dom_insert_apply)
apply(auto simp: zpred_def)
done

```

By the way, the attribute zstrip can also be used to convert a zlemma into a HOL-theorem, such that it is amenable to forward proof:

```

thm lemma5[zstrip]

```

```

po Rel_Refinement.fwRefinementOp_AddBirthday_2

```

To show:

1. \forall BirthdayBook \bullet (\forall BirthdayBook1 \bullet (\forall BirthdayBook1 \bullet (\forall date? name?. pre AddBirthday \wedge Abs \wedge AddBirthday1 \longrightarrow \exists BirthdayBook \bullet Abs' \wedge AddBirthday)))

After structural simplification:

```

apply(zstrip, clarify)
apply(zelim_pre, zintro_sch_ex)

```

...we use the equations to construct the trivial successor state.

```
apply(rule_tac [2] refl)
prefer 2

apply(subgoal_tac "BirthdayBook(birthday',known')")
apply(rotate_tac 1, assumption)
```

We have to prove:

1. $\bigwedge \text{birthday known dates hwm names dates' hwm' names' date? name? birthday' known'}$.
 $\llbracket \text{BirthdayBook (birthday, known); BirthdayBook1 (dates, hwm, names); BirthdayBook1 (dates', hwm', names')} \rrbracket; \text{Abs; AddBirthday1; AddBirthday} \rrbracket \implies \text{BirthdayBook (birthday', known')}$
2. $\bigwedge \text{birthday known dates hwm names dates' hwm' names' date? name? birthday' known'}$.
 $\llbracket \text{BirthdayBook (birthday, known); BirthdayBook1 (dates, hwm, names); BirthdayBook1 (dates', hwm', names')} \rrbracket; \text{Abs; AddBirthday1; AddBirthday} \rrbracket \implies \text{Abs' } \wedge \text{ AddBirthday}$

...which boils down to extracting this knowledge from schema *AddBirthday*:

```
apply(zunfold AddBirthday_def)
apply(tactic "full_expand_schema_tac [thm"AddBirthday_def"] 1")
```

```
apply(drule DECL_D1)
apply(simp add: Z2HOL)
```

Moreover, we have to prove:

1. $\bigwedge \text{birthday known dates hwm names dates' hwm' names' date? name? birthday' known'}$.
 $\llbracket \text{BirthdayBook (birthday, known); BirthdayBook1 (dates, hwm, names); BirthdayBook1 (dates', hwm', names')} \rrbracket; \text{Abs; AddBirthday1; AddBirthday} \rrbracket \implies \text{Abs' } \wedge \text{ AddBirthday}$

The validity of the transition of *AddBirthday* follows from the assumptions.

```
apply simp
```

It remains to show *Abs'*. We unfold several schemas:

```
apply(zunfold (0) Abs_def)
apply(rule DECL_I, unfold Abs_def AddBirthday_def)
apply(drule DECL_D1) back
apply(simp add: set_simps prod_simps Z2HOL)
```

...and achieve

1. $\bigwedge \text{birthday known dates hwm names dates' hwm' names' date? name? birthday' known'}$.
 $\llbracket \text{BirthdayBook (birthday, known); BirthdayBook1 (dates, hwm, names); BirthdayBook1 (dates', hwm', names')} \rrbracket; \text{Abs; AddBirthday1; AddBirthday; BirthdayBook' } \wedge \text{ BirthdayBook1' } \rrbracket \implies \text{known'}$
 $= \text{rel_appl names' } \langle (\lambda i. i \in (1 .. hwm')) \wedge (\forall i \in 1 .. hwm'. \text{birthday}'(.(\text{names}'(.i)).)) = \text{dates}'(.i)) \rangle$

For the latter, we simply apply lemmas 4) and 5).

```
apply(rule conjI)
apply(simp add: image_def asSet_def maplet_def)
apply(zrule lemma4, auto)
apply(zrule lemma5, auto)
discharged
```

4.5 Global Checks

Now we check that all refinement conditions have indeed been proven:

```
check_po except ccOp ccState
```

This concludes the refinement proof based on Forward Simulation in the style of Spivey for the Birthdaybook example.

```
end
```

5 Include both Refinement Proof Alternatives

```
theory BB
imports Fun_Refinement
        Rel_Refinement
```

```
begin
```

```
end
```